課題9の解説

桂田 祐史

2013年7月17日

(1) 661775625 を素因数分解せよ。

(2) $2^{15}-1$ と $2^{20}-1$ の最大公約数を求めよ。

(3) $(a+b)^5$ の展開公式を作れ。

(4) 2次方程式 $x^2+ax+b=0$ を解け。3次方程式 $x^3+px+q=0$ を解け。

(5) 次の関数を微分せよ。(i) $x^2\sqrt{x}+(x^3-x)\sqrt{x^2+x+1}$ (ii) $\sqrt{\dfrac{1+x^2}{1-x^2}}$

(6) (i) $\displaystyle\int_0^1 \frac{1}{(x-2)^5}\,dx$ (ii) $\displaystyle\int_0^\pi \frac{1}{2+\cos x}\,dx$

(1) $n=661775625$ の素因数分解は？

```
n = 661775625

661775625

FactorInteger[n]

{{3,2},{5,4},{7,6}}
```

これは

$$661775625 = 3^2\cdot 5^4\cdot 7^6$$

を意味している。かけ算して n との差を取って確認する。

```
3^2 5^4 7^6 - n

0
```

素因数分解は正しそうです。

(2) 2^{15-1} と $2^{20}-1$ の最大公約数を求める。

GCD[2^15 − 1, 2^20 − 1]

31

31 との答え。素因数分解してみる (普通は最大公約数を求めるより、素因数分解する方がはるかに大変ですが、コンピューターが処理できればそれで構わないわけですね)。

FactorInteger[2^15 − 1]

$\{\{7,1\},\{31,1\},\{151,1\}\}$

FactorInteger[2^20 − 1]

$\{\{3,1\},\{5,2\},\{11,1\},\{31,1\},\{41,1\}\}$

これは

$$2^{15}-1 = 7^1\cdot 31^1\cdot 151^1,\quad 2^{20}-1 = 3^1\cdot 5^2\cdot 11^1\cdot 31^1\cdot 41^1$$

を意味しています。この素因数分解を認めれば (確かめる気になれば積を計算すれば良いでしょう) 確かに最大公約数は 31 です。

(3) $(a+b)^5$ を展開せよ。これは簡単で,

Expand[($a+b$)^5]

$a^5+5a^4b+10a^3b^2+10a^2b^3+5ab^4+b^5$

これは

$$(a+b)^5 = a^5+5a^4b+10a^3b^2+10a^2b^3+5ab^4+b^5$$

ということです。簡単化と，2 項係数 ${}_5\mathrm{C}_r$ $(r=0,1,\ldots,5)$ を計算して見ます。

Simplify[%]

$(a+b)^5$

Table[Binomial[5, r], {r, 0, 5}]

$\{1,5,10,10,5,1\}$

この場合、簡単化は因数分解することだったようです。一方、係数が確かに 2 項係数であることも分かります。

(4) $x^2+ax+b=0$ を解け。

sol = Solve[x^2 + ax + b == 0, x]

$\left\{\left\{x\to \frac{1}{2}\left(-a-\sqrt{a^2-4b}\right)\right\},\left\{x\to \frac{1}{2}\left(-a+\sqrt{a^2-4b}\right)\right\}\right\}$

これは

$$x=\frac{1}{2}\left(-a-\sqrt{a^2-4b}\right),\frac{1}{2}\left(-a+\sqrt{a^2-4b}\right)$$

を意味しています。2 次方程式の解の公式を叩き込まれた人は、目で検算できるでしょうが、代入チェックをしてみます。

x^2 + ax + b/.sol

$\left\{\frac{1}{2}a\left(-a-\sqrt{a^2-4b}\right)+\frac{1}{4}\left(-a-\sqrt{a^2-4b}\right)^2+b,\frac{1}{2}a\left(-a+\sqrt{a^2-4b}\right)+\frac{1}{4}\left(-a+\sqrt{a^2-4b}\right)^2+b\right\}$

Simplify[%]

$\{0,0\}$

大丈夫そうですね。

「$x^3+px+q=0$ を解け」。

sol = Solve[x^3 + px + q == 0, x]

$$\left\{\left\{x \to -\frac{\left(\frac{2}{3}\right)^{1/3} p}{\left(-9q+\sqrt{3}\sqrt{4p^3+27q^2}\right)^{1/3}} + \frac{\left(-9q+\sqrt{3}\sqrt{4p^3+27q^2}\right)^{1/3}}{2^{1/3}3^{2/3}}\right\}, \left\{x \to \frac{\left(1+i\sqrt{3}\right)p}{2^{2/3}3^{1/3}\left(-9q+\sqrt{3}\sqrt{4p^3+27q^2}\right)^{1/3}} - \frac{\left(1-i\sqrt{3}\right.}{}\right.\right.$$

例年、「長くて 1 行に入りません」という SOS があります。以下では、`x1`, `x2`, `x3` に代入して，一つずつ表示しています。

{x1, x2, x3} = x/.sol;

x1

$$-\frac{\left(\frac{2}{3}\right)^{1/3} p}{\left(-9q+\sqrt{3}\sqrt{4p^3+27q^2}\right)^{1/3}} + \frac{\left(-9q+\sqrt{3}\sqrt{4p^3+27q^2}\right)^{1/3}}{2^{1/3}3^{2/3}}$$

x2

$$\frac{\left(1+i\sqrt{3}\right)p}{2^{2/3}3^{1/3}\left(-9q+\sqrt{3}\sqrt{4p^3+27q^2}\right)^{1/3}} - \frac{\left(1-i\sqrt{3}\right)\left(-9q+\sqrt{3}\sqrt{4p^3+27q^2}\right)^{1/3}}{2\ 2^{1/3}3^{2/3}}$$

x3

$$\frac{\left(1-i\sqrt{3}\right)p}{2^{2/3}3^{1/3}\left(-9q+\sqrt{3}\sqrt{4p^3+27q^2}\right)^{1/3}} - \frac{\left(1+i\sqrt{3}\right)\left(-9q+\sqrt{3}\sqrt{4p^3+27q^2}\right)^{1/3}}{2\ 2^{1/3}3^{2/3}}$$

代入して簡単化して 0 になるかチェックしてみます。

x^3 + px + q/.sol

$$\left\{q + p\left(-\frac{\left(\frac{2}{3}\right)^{1/3} p}{\left(-9q+\sqrt{3}\sqrt{4p^3+27q^2}\right)^{1/3}} + \frac{\left(-9q+\sqrt{3}\sqrt{4p^3+27q^2}\right)^{1/3}}{2^{1/3}3^{2/3}}\right) + \left(-\frac{\left(\frac{2}{3}\right)^{1/3} p}{\left(-9q+\sqrt{3}\sqrt{4p^3+27q^2}\right)^{1/3}} + \frac{\left(-9q+\sqrt{3}\sqrt{4p^3+\right.}{2^{1/3}3^{2/3}}\right.\right.$$

Simplify[%]

$\{0, 0, 0\}$

大丈夫そう。

結果は一見ぐちゃぐちゃに見えますが、共通部分が多いことに気づきます。

$$A := -9q + \sqrt{3}\sqrt{4p^3 + 27q^2}$$

とおくと、簡単になりそう。また Cardano の公式をうっすらとでも知っていれば

$$\omega = \frac{-1+\sqrt{3}i}{2}, \quad \omega^2 = \frac{-1-\sqrt{3}i}{2}$$

を使うと見やすくなりそう、と見当がつく。そこで

$$\mathbf{sol/.-9}\boldsymbol{q}\mathbf{+Sqrt[3]Sqrt[4}\boldsymbol{p}\mathbf{\hat{}3+27}\boldsymbol{q}\mathbf{\hat{}2]\to}\boldsymbol{A}$$

$$\left\{\left\{x \to \frac{A^{1/3}}{2^{1/3}3^{2/3}} - \frac{\left(\frac{2}{3}\right)^{1/3}p}{A^{1/3}}\right\}, \left\{x \to -\frac{\left(1-i\sqrt{3}\right)A^{1/3}}{2\ 2^{1/3}3^{2/3}} + \frac{\left(1+i\sqrt{3}\right)p}{2^{2/3}3^{1/3}A^{1/3}}\right\}, \left\{x \to -\frac{\left(1+i\sqrt{3}\right)A^{1/3}}{2\ 2^{1/3}3^{2/3}} + \frac{\left(1-i\sqrt{3}\right)p}{2^{2/3}3^{1/3}A^{1/3}}\right\}\right\}$$

$$\mathbf{\%/.\{1+}\boldsymbol{I}\mathbf{Sqrt[3]\to -2om2, 1-}\boldsymbol{I}\mathbf{Sqrt[3]\to -2om\}}$$

$$\left\{\left\{x \to \frac{A^{1/3}}{2^{1/3}3^{2/3}} - \frac{\left(\frac{2}{3}\right)^{1/3}p}{A^{1/3}}\right\}, \left\{x \to \frac{A^{1/3}\mathrm{om}}{2^{1/3}3^{2/3}} - \frac{\left(\frac{2}{3}\right)^{1/3}\mathrm{om2}p}{A^{1/3}}\right\}, \left\{x \to \frac{A^{1/3}\mathrm{om2}}{2^{1/3}3^{2/3}} - \frac{\left(\frac{2}{3}\right)^{1/3}\mathrm{om}p}{A^{1/3}}\right\}\right\}$$

Cardano の公式を知ってのお化粧をしてみます。

$$u := \frac{A}{18} = -\frac{q}{2} + \sqrt{\frac{p^3}{27} + \frac{q^2}{4}}$$

とおいてみると

$$\mathbf{\%/.}\boldsymbol{A}\mathbf{\to 18}\boldsymbol{U}$$

$$\left\{\left\{x \to -\frac{p}{3U^{1/3}} + U^{1/3}\right\}, \left\{x \to -\frac{\mathrm{om2}p}{3U^{1/3}} + \mathrm{om}U^{1/3}\right\}, \left\{x \to -\frac{\mathrm{om}p}{3U^{1/3}} + \mathrm{om2}U^{1/3}\right\}\right\}$$

つまり

$$x = -\frac{p}{3U^{1/3}} + U^{1/3}, -\frac{p}{3U^{1/3}}\omega^2 + U^{1/3}\omega, -\frac{p}{3U^{1/3}}\omega + U^{1/3}\omega^2,$$

ということです。次に掲げる Cardano の公式と見比べてみましょう。

Cardano の公式 (少し変なまとめ方)

$x^3+px+q=0$ の解は、

$$u := \sqrt[3]{-\frac{q}{2}+\sqrt{\frac{p^3}{27}+\frac{q^2}{4}}},$$

$$v := \frac{-p/3}{u}$$

とおくと、

$$x = u+v, u\omega+v\omega^2, u\omega^2+v\omega.$$

無事、一致していることが分ります。

(5) (i) $f(x) := x^2\sqrt{x}+(x^3-x)\sqrt{x^2+x+1}$ の導関数を計算せよ。

```
f[x_]:=x^2Sqrt[x] + (x^3 - x)Sqrt[x^2 + x + 1]

D[f[x], x]
```

$$\frac{5x^{3/2}}{2}+\sqrt{1+x+x^2}\left(-1+3x^2\right)+\frac{(1+2x)\left(-x+x^3\right)}{2\sqrt{1+x+x^2}}$$

```
Simplify[%]
```

$$\frac{-2-3x+2x^2+7x^3+8x^4+5x^{3/2}\sqrt{1+x+x^2}}{2\sqrt{1+x+x^2}}$$

これは

$$f'(x) = \frac{5}{2}x^{3/2}+\frac{-2-3x+2x^2+7x^3+8x^4}{2\sqrt{1+x+x^2}}$$

ということを意味しています。

安直だが，積分して，引き算してみます (上の素因数分解と同様、普通は積分の方が大変なので、こんな検算はしませんが…)。定数になれば正しいはずです。

Integrate[%, x]

$x\left(x^{3/2}+(-1+x^2)\sqrt{1+x+x^2}\right)$

% − $f[x]$

$-x^{5/2}-\sqrt{1+x+x^2}\,(-x+x^3)+x\left(x^{3/2}+(-1+x^2)\sqrt{1+x+x^2}\right)$

Simplify[%]

0

0 になったので大丈夫そうです。

別のチェックとしては、数値微分してグラフを比較するというのが考えられます。これはこれで一つのやり方だと思います。

$D[f[x], x]$

$\frac{5x^{3/2}}{2}+\sqrt{1+x+x^2}\,(-1+3x^2)+\frac{(1+2x)\left(-x+x^3\right)}{2\sqrt{1+x+x^2}}$

Simplify[%]

$\frac{-2-3x+2x^2+7x^3+8x^4+5x^{3/2}\sqrt{1+x+x^2}}{2\sqrt{1+x+x^2}}$

g1 = Plot[%, {x, 0, 2}]

g2 = Plot[($f[x+0.001]-f[x]$)/0.001, {x, 0, 2}]

図 1, 2 を見比べて下さい。目で見る限りの精度 (これは案外低いですが) で、一致しているように見えます。

(ii) 今度は $f(x)=\sqrt{\dfrac{1+x^2}{1-x^2}}$ の導関数です。



図 1: 導関数のグラフ



図 2: 数値微分で得た関数のグラフ

```
f[x_]:=Sqrt[(1 + x^2)/(1 − x^2)]

f′[x]
```

$$\frac{\frac{2x}{1-x^2}+\frac{2x\left(1+x^2\right)}{\left(1-x^2\right)^2}}{2\sqrt{\frac{1+x^2}{1-x^2}}}$$

```
Simplify[%]
```

$$\frac{2x}{\left(-1+x^2\right)^2\sqrt{\frac{1+x^2}{1-x^2}}}$$

```
Integrate[%, x]
```

$$\sqrt{\frac{1+x^2}{1-x^2}}$$

$$\left(\sqrt{\frac{1+x^2}{1-x^2}}\right)' = \frac{2x}{(1-x^2)^2\sqrt{\frac{1+x^2}{1-x^2}}} = \frac{2x}{(1-x^2)\sqrt{1-x^4}}.$$

この問題で数値微分してグラフを比較するのはお任せします。

(6) (i) $\displaystyle\int_0^1 \frac{1}{(x-2)^5}dx$ を求める。

```
f[x_]:=1/(x - 2)^5

Integrate[f[x], {x, 0, 1}]
```

$-\frac{15}{64}$

$$\int_0^1 \frac{1}{(x-2)^5} dx = -\frac{15}{64}.$$

この定積分をチェックするため，不定積分から計算してみます。また不定積分を微分してもとの関数と一致するか調べます。

```
y = Integrate[f[x], x]
```

$-\frac{1}{4(-2+x)^4}$

```
(y/.x → 1) - (y/.x → 0)
```

$-\frac{15}{64}$

```
D[y, x]
```

$\frac{1}{(-2+x)^5}$

不定積分を使って定積分を計算して一致しました。また不定積分を微分すると元に戻ることが確認できました。他に数値積分して比較するという手もあります。

(ii) $\displaystyle\int_0^{\pi} \frac{1}{2+\cos x} dx$ を求める。

$f[x\_]{:=}1/(2+\mathrm{Cos}[x])$

$\mathrm{Integrate}[f[x],\{x,0,\mathrm{Pi}\}]$

$\frac{\pi}{\sqrt{3}}$

$$\int_0^{\pi} \frac{1}{2+\cos x}dx = \frac{\pi}{\sqrt{3}}.$$

$\mathrm{Integrate}[f[x],x]$

$\frac{2\mathrm{ArcTan}\left[\frac{\mathrm{Tan}\left[\frac{x}{2}\right]}{\sqrt{3}}\right]}{\sqrt{3}}$

$\mathrm{Simplify}[\%]$

$\frac{2\mathrm{ArcTan}\left[\frac{\mathrm{Tan}\left[\frac{x}{2}\right]}{\sqrt{3}}\right]}{\sqrt{3}}$

$D[\%,x]$

$\frac{\mathrm{Sec}\left[\frac{x}{2}\right]^2}{3\left(1+\frac{1}{3}\mathrm{Tan}\left[\frac{x}{2}\right]^2\right)}$

$\% - f[x]$

$-\frac{1}{2+\mathrm{Cos}[x]} + \frac{\mathrm{Sec}\left[\frac{x}{2}\right]^2}{3\left(1+\frac{1}{3}\mathrm{Tan}\left[\frac{x}{2}\right]^2\right)}$

$\mathrm{Simplify}[\%]$

0

この問題には、ちょっと悩ましいところがあります (どうしてでしょう？わざとぼかします)。うまく検算できたら、レポート課題 11 として認めても良いかも (甘いかな？)。